
Overview: fzxNGN is a 2D physics engine library that was ported to 
QB64 from the Impulse engine written by Randy Gaul 
https://github.com/RandyGaul/ImpulseEngine.

Features: 
• Rigid body simulation
• Circle and polygon primitives, concave objects not supported yet
• Joint simulation
• Camera Library
• Input Library
• Finite State Machine helper functions
• Perlin noise library functions
• XML parsing (WIP)
• LERP functions
• FPS helper functions
• Tons of vector and matrix math functions.
• Units are arbitrary, Its up to the user to decide which units of

length to use.

What it is and what it is not:
• A project to help QB64 programmers such as myself make more 

interesting demos and mini-games.
• It is a collection of subroutines and functions that I’ve made 

every attempt to generalize. They can be used outside the 
simulation.

• Not a core engine for the next AAA game.
• Not for serious engineering use.

https://github.com/RandyGaul/ImpulseEngine


Compiling examples:

• Compiling should be straightforward. Load the example file 
and hit F5 or Start. 

• Make sure the “Output EXE to Source Folder” is selected. This
will ensure that the file structure is not broken.



FzxNGN File structure:
• The filenames may vary but the same general file structure 

should be maintained.



FzxNGN Globals (fzxNGN_ini.bas):

__fzxBody(): Contains all the data pertaining to each rigid body.
__fzxJoints(): Contains  all the data pertaining to the joints.
__fzxHits(): Collision information
__fzxCamera : Camera data
__fzxWorld : World data
__fzxFPSCount : FPS counting
__fzxInputDevice : Mouse and keyboard
__fzxSettings : Generalized settings. (Currently only mouse double click timing)



A Bare Bone Implementation:

This is a simple example of what it takes for a small implementation 
of the engine.

• Initialization include
◦ '$include:'..\fzxNGN_BASE_v2\fzxNGN_ini.bas'
◦ sets up the types(UDT), global variables, and constants

• Call to Build scene.  
• Main Loop

◦ Call to Animate scene
▪ Player interaction happens here

◦ Call to fzxNGN to calculate the next step
▪ fzxImpulseStep delta time, iterations

• Delta Time is your time step i.e. 1/30 of a second
• Iterations how many steps the simulation runs per step.

◦ Call to Render scene

• Include core code
◦ '$include:'..\fzxNGN_BASE_v2\fzxNGN_BASE.bas'
◦ All the core functionality is contained here

• Build Scene
◦ Initial setup of the camera. 
◦ Set some limits on the world. 
◦ Setup gravity.
◦ Add your bodies to the simulation.

• Animate Scene
◦ This where you will interact with the bodies.

• Render Scene
◦ Draw the bodies in the scene.



Code Example: fzxNGNBareBones.bas

• '$DYNAMIC
◦ Some of the Global variable are dynamic and this must be declared

• '$include:'..\fzxNGN_BASE_v2\fzxNGN_ini.bas'
◦ Include the types, constants, and the global variables

• SCREEN _NEWIMAGE(1024, 768, 32)
◦ Screen size can be whatever you desire, because the camera functionality will allow the user

to view any portion of the world.
• DIM AS LONG iterations: iterations = 2

◦ Number of simulation steps ran before rendering. 
• DIM SHARED AS DOUBLE dt: dt = 1 / 60

◦ Time step of simulation.

'$DYNAMIC
'$include:'..\fzxNGN_BASE_v2\fzxNGN_ini.bas'
SCREEN _NEWIMAGE(1024, 768, 32)
DIM AS LONG iterations: iterations = 2
DIM SHARED AS DOUBLE dt: dt = 1 / 60



• buildScene
◦ Call to the subroutine that will setup the world and build your initial scene

• DO
◦ This is the Main loop of the program

•   CLS: LOCATE 1: PRINT "Click the mouse on the play field to spawn an object"
•   fzxHandleInputDevice 

◦ A subroutine that will manage some extra routines for the mouse and keyboard
◦ Later in the program, “__fzxInputDevice.mouse.b1.NegEdge” flag will be queried, It will 

return a true if the mouse button has been released.
•   Animatescene

◦ This is a call to the subroutine that handles all of the frame to frame activities.
◦ User interaction
◦ Level animation and logic

•   fzxImpulseStep dt, iterations 
◦ Run the simulation

•   renderBodies
◦ Draw the level
◦ This is up to the user to implement depending on their need.

• _DISPLAY
• LOOP UNTIL INKEY$ = CHR$(27)
• SYSTEM
• '$include:'..\fzxNGN_BASE_v2\fzxNGN_BASE.bas'

◦ Provides all of the fzxNGN functionality

buildScene
DO
  CLS: LOCATE 1: PRINT "Click the mouse on the play field to spawn an object"
  fzxHandleInputDevice 
  animatescene
  fzxImpulseStep dt, iterations 
  renderBodies
_DISPLAY
LOOP UNTIL INKEY$ = CHR$(27)
SYSTEM
'$include:'..\fzxNGN_BASE_v2\fzxNGN_BASE.bas'



• SUB animateScene
•   DIM AS LONG temp

◦ temp is a variable we will use, because after the body is setup, we don’t care about it 
anymore.

• IF __fzxInputDevice.mouse.b1.NegEdge THEN 
◦ This condition will be true when the user releases the mouse button

•     IF RND > .5 THEN
◦ Flip a coin

•       temp = fzxCreateCircleBodyEx("b" + _TRIM$(STR$(RND * 1000000000)), 10)
◦ Create a circle body with some unique name with a radius of 10.

•     ELSE
•       temp = fzxCreateBoxBodyEx("b" + _TRIM$(STR$(RND * 1000000000)), 10, 10)

◦ Create a box body with as unique name with a dimension of 10 by 10.
• END IF
• fzxSetBody cFZX_PARAMETER_POSITION, temp, __fzxInputDevice.mouse.worldPosition.x, 

__fzxInputDevice.mouse.worldPosition.y
◦ Move the previously created body to the position in the game world pointed to by the mouse.
◦  “__fzxInputDevice.mouse.worldPosition” is not the same as screen position, it calculated 

based on camera position and camera zoom.
• fzxSetBody cFZX_PARAMETER_VELOCITY, temp, __fzxInputDevice.mouse.velocity.x, 

__fzxInputDevice.mouse.velocity.y
◦ Give the body some velocity based on how fast the mouse was moving when the user released 

the mouse button. 

SUB animateScene
  DIM AS LONG temp
  IF __fzxInputDevice.mouse.b1.NegEdge THEN 
    IF RND > .5 THEN
      temp = fzxCreateCircleBodyEx("b" + _TRIM$(STR$(RND * 1000000000)), 10)
    ELSE
      temp = fzxCreateBoxBodyEx("b" + _TRIM$(STR$(RND * 1000000000)), 10, 10)
    END IF

    fzxSetBody cFZX_PARAMETER_POSITION, temp, __fzxInputDevice.mouse.worldPosition.x, 
__fzxInputDevice.mouse.worldPosition.y

    fzxSetBody cFZX_PARAMETER_VELOCITY, temp, __fzxInputDevice.mouse.velocity.x, 
__fzxInputDevice.mouse.velocity.y

    fzxSetBody cFZX_PARAMETER_ORIENT, temp, _D2R(RND * 360), 0
    fzxSetBody cFZX_PARAMETER_RESTITUTION, temp, .5, 0 ' Bounce
    fzxSetBody cFZX_PARAMETER_STATICFRICTION, temp, .1, 0
    fzxSetBody cFZX_PARAMETER_DYNAMICFRICTION, temp, .85, 0
    fzxSetBody cFZX_PARAMETER_LIFETIME, temp, RND * 20 + 10, 0
  END IF
END SUB



• fzxSetBody cFZX_PARAMETER_ORIENT, temp, _D2R(RND * 360), 0
◦ Give the body an arbitrary angle

• fzxSetBody cFZX_PARAMETER_RESTITUTION, temp, .5, 0 ' Bounce
◦ Set the bounce of the body. 
◦ Value should be between 0 to 1. Zero is no bounce at all and one is a very hyper super ball.

• fzxSetBody cFZX_PARAMETER_STATICFRICTION, temp, .1, 0
• fzxSetBody cFZX_PARAMETER_DYNAMICFRICTION, temp, .85, 0

◦ Static and dynamic(kinetic) friction can be best described in this article on wikipedia.
◦ https://en.wikipedia.org/wiki/Friction  

• fzxSetBody cFZX_PARAMETER_LIFETIME, temp, RND * 20 + 10, 0
◦ Delete the body after a random number of seconds.

•   END IF
• END SUB

https://en.wikipedia.org/wiki/Friction


• SUB buildScene
•   DIM AS LONG temp

◦ We will be using this to create static objects that we wont need later 
•   __fzxCamera.zoom = 1
•   fzxCalculateFOV

◦ Set up Camera zoom. Note “fzxCalculateFOV” needs to be called every time the zoom is 
changed.

•   fzxVector2DSet __fzxWorld.minusLimit, -200000, -200000
•   fzxVector2DSet __fzxWorld.plusLimit, 200000, 200000

◦ Set world limits. Objects outside of this are will be deleted.
•   fzxVector2DSet __fzxWorld.spawn, 0, 0

◦ This is a position that can be used how the user likes, but I use it to set the starting 
position of everything

•   fzxVector2DSet __fzxWorld.gravity, 0.0, 10.0
◦ Set the gravity vector

•   fzxVector2DSet __fzxCamera.position, __fzxWorld.spawn.x, __fzxWorld.spawn.y – 300
◦ Set the camera position

•   DIM o AS tFZX_VECTOR2d
•   fzxVector2DMultiplyScalarND o, __fzxWorld.gravity, dt
•   __fzxWorld.resting = fzxVector2DLengthSq(o) + cFZX_EPSILON

◦ Setting up some simulation related values
◦ ToDo: Do this automatically

•   temp = fzxCreateBoxBodyEx("floor", 800, 10)
◦ Create a floor body

•   fzxSetBody cFZX_PARAMETER_POSITION, temp, __fzxWorld.spawn.x, __fzxWorld.spawn.y
◦ Set it at the spawn point

•   fzxSetBody cFZX_PARAMETER_STATIC, temp, 0, 0
◦ Set it as a static object

• END SUB

SUB buildScene
  DIM AS LONG temp
  __fzxCamera.zoom = 1
  fzxCalculateFOV
  fzxVector2DSet __fzxWorld.minusLimit, -200000, -200000
  fzxVector2DSet __fzxWorld.plusLimit, 200000, 200000
  fzxVector2DSet __fzxWorld.spawn, 0, 0
  fzxVector2DSet __fzxWorld.gravity, 0.0, 10.0
  fzxVector2DSet __fzxCamera.position, __fzxWorld.spawn.x, __fzxWorld.spawn.y - 300
  DIM o AS tFZX_VECTOR2d
  fzxVector2DMultiplyScalarND o, __fzxWorld.gravity, dt
  __fzxWorld.resting = fzxVector2DLengthSq(o) + cFZX_EPSILON
  temp = fzxCreateBoxBodyEx("floor", 800, 10)
  fzxSetBody cFZX_PARAMETER_POSITION, temp, __fzxWorld.spawn.x, __fzxWorld.spawn.y
  fzxSetBody cFZX_PARAMETER_STATIC, temp, 0, 0
END SUB



• SUB renderBodies STATIC
•   DIM i AS LONG
•   DIM AS tFZX_VECTOR2d scSize, scMid, scUpperLeft, camUpperLeft, aabbUpperLeft, aabbSize, 

aabbHalfSize
•   DIM AS LONG ub: ub = UBOUND(__fzxBody)
•   fzxVector2DSet aabbSize, 40000, 40000
•   fzxVector2DSet aabbHalfSize, aabbSize.x / 2, aabbSize.y / 2
•   fzxVector2DSet scUpperLeft, 0, 0
•   fzxVector2DSet scSize, _WIDTH, _HEIGHT
•   fzxVector2DDivideScalarND scMid, scSize, 2
•   fzxVector2DSubVectorND camUpperLeft, __fzxCamera.position, scMid
•   i = 0: DO WHILE i < ub
•     IF __fzxBody(i).enable THEN
•       'fzxAABB to cut down on rendering objects out of camera view
•       fzxVector2DSubVectorND aabbUpperLeft, __fzxBody(i).fzx.position, aabbHalfSize
•       IF fzxAABBOverlap(camUpperLeft.x, camUpperLeft.y, scSize.x, scSize.y, aabbUpperLeft.x, 

aabbUpperLeft.y, aabbSize.x, aabbSize.y) THEN
•         IF __fzxBody(i).shape.ty = cFZX_SHAPE_CIRCLE THEN
•           renderWireFrameCircle i, _RGB32(0, 255, 0)
•         ELSE IF __fzxBody(i).shape.ty = cFZX_SHAPE_POLYGON THEN
•             renderWireFramePoly i

SUB renderBodies STATIC
  DIM i AS LONG
  DIM AS tFZX_VECTOR2d scSize, scMid, scUpperLeft, camUpperLeft, aabbUpperLeft, aabbSize, aabbHalfSize
  DIM AS LONG ub: ub = UBOUND(__fzxBody)

  fzxVector2DSet aabbSize, 40000, 40000
  fzxVector2DSet aabbHalfSize, aabbSize.x / 2, aabbSize.y / 2

  fzxVector2DSet scUpperLeft, 0, 0
  fzxVector2DSet scSize, _WIDTH, _HEIGHT

  fzxVector2DDivideScalarND scMid, scSize, 2
  fzxVector2DSubVectorND camUpperLeft, __fzxCamera.position, scMid

  i = 0: DO WHILE i < ub
    IF __fzxBody(i).enable THEN
      'fzxAABB to cut down on rendering objects out of camera view
      fzxVector2DSubVectorND aabbUpperLeft, __fzxBody(i).fzx.position, aabbHalfSize
      IF fzxAABBOverlap(camUpperLeft.x, camUpperLeft.y, scSize.x, scSize.y, aabbUpperLeft.x, 
aabbUpperLeft.y, aabbSize.x, aabbSize.y) THEN
        IF __fzxBody(i).shape.ty = cFZX_SHAPE_CIRCLE THEN
          renderWireFrameCircle i, _RGB32(0, 255, 0)
        ELSE IF __fzxBody(i).shape.ty = cFZX_SHAPE_POLYGON THEN
            renderWireFramePoly i
          END IF
        END IF
      END IF
    END IF
    i = i + 1
  LOOP
END SUB



•           END IF
•         END IF
•       END IF
•     END IF
•     i = i + 1
•   LOOP
• END SUB



• SUB renderWireFrameCircle (index AS LONG, c AS LONG)
•   DIM AS tFZX_VECTOR2d o1, o2
•   fzxWorldToCameraEx __fzxBody(index).fzx.position, o1
•   CIRCLE (o1.x, o1.y), __fzxBody(index).shape.radius * __fzxCamera.zoom, c
•   o2.x = o1.x + (__fzxBody(index).shape.radius * __fzxCamera.zoom) * 

COS(__fzxBody(index).fzx.orient)
•   o2.y = o1.y + (__fzxBody(index).shape.radius * __fzxCamera.zoom) * 

SIN(__fzxBody(index).fzx.orient)
•   LINE (o1.x, o1.y)-(o2.x, o2.y), c
• END SUB

SUB renderWireFrameCircle (index AS LONG, c AS LONG)
  DIM AS tFZX_VECTOR2d o1, o2
  fzxWorldToCameraEx __fzxBody(index).fzx.position, o1
  CIRCLE (o1.x, o1.y), __fzxBody(index).shape.radius * __fzxCamera.zoom, c
  o2.x = o1.x + (__fzxBody(index).shape.radius * __fzxCamera.zoom) * COS(__fzxBody(index).fzx.orient)
  o2.y = o1.y + (__fzxBody(index).shape.radius * __fzxCamera.zoom) * SIN(__fzxBody(index).fzx.orient)
  LINE (o1.x, o1.y)-(o2.x, o2.y), c
END SUB



• SUB renderWireFramePoly (index AS LONG)
•   DIM vert(3) AS tFZX_VECTOR2d
•

•   fzxGetBodyVert index, 0, vert(0)
•   fzxWorldToCamera index, vert(0)
•

•   fzxGetBodyVert index, 1, vert(1)
•   fzxWorldToCamera index, vert(1)
•

•   fzxGetBodyVert index, 2, vert(2)
•   fzxWorldToCamera index, vert(2)
•

•   fzxGetBodyVert index, 3, vert(3)
•   fzxWorldToCamera index, vert(3)
•

•   LINE (vert(0).x, vert(0).y)-(vert(1).x, vert(1).y), _RGB(0, 255, 0)
•   LINE (vert(1).x, vert(1).y)-(vert(2).x, vert(2).y), _RGB(0, 255, 0)
•   LINE (vert(2).x, vert(2).y)-(vert(3).x, vert(3).y), _RGB(0, 255, 0)
•   LINE (vert(3).x, vert(3).y)-(vert(0).x, vert(0).y), _RGB(0, 255, 0)
• END SUB

SUB renderWireFramePoly (index AS LONG)
  DIM vert(3) AS tFZX_VECTOR2d

  fzxGetBodyVert index, 0, vert(0)
  fzxWorldToCamera index, vert(0)

  fzxGetBodyVert index, 1, vert(1)
  fzxWorldToCamera index, vert(1)

  fzxGetBodyVert index, 2, vert(2)
  fzxWorldToCamera index, vert(2)

  fzxGetBodyVert index, 3, vert(3)
  fzxWorldToCamera index, vert(3)

  LINE (vert(0).x, vert(0).y)-(vert(1).x, vert(1).y), _RGB(0, 255, 0)
  LINE (vert(1).x, vert(1).y)-(vert(2).x, vert(2).y), _RGB(0, 255, 0)
  LINE (vert(2).x, vert(2).y)-(vert(3).x, vert(3).y), _RGB(0, 255, 0)
  LINE (vert(3).x, vert(3).y)-(vert(0).x, vert(0).y), _RGB(0, 255, 0)
END SUB



Body Creation:

The following function create bodies for the simulation.

fzxCreateCircleBodyEx (“unique name for object”, radius)
• Adds a circle body to simulation
• returns an index to the body in the __fzxBody() array

fzxCreateBoxBodyEx (“unique name for object”, Width, Height)
• Adds a Rectangle to the simulation
• returns an index to the body in the __fzxBody() array

fzxCreateTrapBodyEx (“unique name for object”, Width, Height, yoff1, yoff2)
• Adds a trapezoid to the simulation
• returns an index to the body in the __fzxBody() array

An example of a body creation would be like:

1.   box = fzxCreateBoxBodyEx("box", 100, 100)
2.   fzxSetBody cFZX_PARAMETER_POSITION, box, 100, 100
3.   fzxSetBody cFZX_PARAMETER_STATIC, box, 0, 0
4.   fzxSetBodyEx cFZX_PARAMETER_ORIENT, "box", _D2R(90), 0

Line 1 creates a box named “box” that is 100 units long by 100 units wide. As 
stated earlier units are arbitrary, so it can be 100 miles or 100 millimeters. Its 
up to the user to decide.

Line 2 the body is moved to a position of 100, 100. Again units are arbitrary.

Line 3 the body is set to static, and acts as a wall or a solid obstacle. You can 
still move it or arrange it as you see fit.

Line 4 the body is now addressed by its name instead of index and the orientation 
is set to 90 degrees.

yoff1

width

height

yoff2



Body Parameters:

The following parameters can be set by the fzxSetBody subroutine. 
• fzxSetBody (Parameter, Index, argument 1, argument 2)

◦ Index in the body in the __fzxBody() array you are changing
◦ The arguments are the new values.

▪ Argument 2 may not always be necessary. Just leave it 0. 
◦ Parameter Contants

• cFZX_PARAMETER_POSITION
◦ Argument 1 – X position in the world
◦ Argument 2 – Y position in the world

• cFZX_PARAMETER_VELOCITY
◦ Argument 1 – X velocity in the world
◦ Argument 2 – Y velocity in the world

• cFZX_PARAMETER_FORCE
◦ Argument 1 – X force applied to body
◦ Argument 2 – Y force applied to body

• cFZX_PARAMETER_ANGULARVELOCITY
◦ Argument 1 – angular velocity in the world
◦ Argument 2 – not used

• cFZX_PARAMETER_TORQUE
◦ Argument 1 – torque force applied to body
◦ Argument 2 – not used

• cFZX_PARAMETER_ORIENT
◦ Argument 1 – body angle in radians
◦ Argument 2 – not used

• cFZX_PARAMETER_STATICFRICTION
◦ Argument 1 – static friction on the body surface
◦ Argument 2 – not used 
◦ More info https://en.wikipedia.org/wiki/Friction

• cFZX_PARAMETER_DYNAMICFRICTION
◦ Argument 1 – dynamic/kinetic friction on the body surface
◦ Argument 2 – not used 
◦ More info https://en.wikipedia.org/wiki/Friction

• cFZX_PARAMETER_RESTITUTION
◦ Argument 1 – bounciness of the body surface
◦ Argument 2 – not used

• cFZX_PARAMETER_COLOR
◦ Argument 1 – color used in wire frame, depends on renderer to 

implement.
◦ Argument 2 – not used

https://en.wikipedia.org/wiki/Friction
https://en.wikipedia.org/wiki/Friction


• cFZX_PARAMETER_ENABLE
◦ Argument 1 – 0 or non zero

▪ Removes body from simulation
▪ can be reenabled

◦ Argument 2 – not used
• cFZX_PARAMETER_STATIC

◦ Sets the object as static and object act like a wall or permanant 
fixture

◦ Argument 1 – not used
◦ Argument 2 – not used

• cFZX_PARAMETER_TEXTURE
◦ Sets the texture for the body, depends on renderer to implement.
◦ Argument 1 – valid texture handle.
◦ Argument 2 – not used

• cFZX_PARAMETER_FLIPTEXTURE 
◦ Flip texture flag, depends on renderer to implement.
◦ Argument 1 – 0 or non zero
◦ Argument 2 – not used

• cFZX_PARAMETER_SCALETEXTURE
◦ Scale texture multiplier, depends on renderer to implement.
◦ Argument 1 – X axis, positive non zero number 
◦ Argument 2 – Y axis, positive non zero number

• cFZX_PARAMETER_OFFSETTEXTURE
◦ Shift texture by offset, depends on renderer to implement.
◦ Argument 1 – X axis
◦ Argument 2 – Y axis

• cFZX_PARAMETER_COLLISIONMASK
◦ Used to selectively allow collisions between bodies

▪ A value of &B00000001 on one body and value &B00000001 on another 
body will collide.

▪ A value of &B00000010 on one body and &B00000001 on another body 
will not collide.

▪ The default is &B11111111.
▪ They essentially logically ANDed together.

◦ Argument 1 – unsigned integer
◦ Argument 2 – not used

• cFZX_PARAMETER_INVERTNORMALS
◦ Experimental feature (I don’t recommend using it)
◦ Argument 1 – unsigned integer
◦ Argument 2 – not used

• cFZX_PARAMETER_NOPHYSICS
◦ Used for sensors. Similar to cFZX_PARAMETER_ENABLE, but body still 

picks up collisions, but wont react to them.
◦ Argument 1 – 0 or non zero
◦ Argument 2 – not used



• cFZX_PARAMETER_SPECIALFUNCTION
◦ User functionality, can be used for whatever the user needs
◦ Argument 1 – any value
◦ Argument 2 – any value

• cFZX_PARAMETER_RENDERORDER
◦ Depreciated – left for compatibility
◦ Argument 1 – any value
◦ Argument 2 – unused

• cFZX_PARAMETER_ENTITYID
◦ User functionality, can be used for whatever the user needs
◦ Argument 1 – any value
◦ Argument 2 – unused

• cFZX_PARAMETER_LIFETIME
◦ Give the body a finite lifetime
◦ Argument 1 – time in seconds
◦ Argument 2 – unused

• cFZX_PARAMETER_REPEATTEXTURE
◦ Repeat texture multiplier, depends on renderer to implement.
◦ Argument 1 – X axis, positive non zero number 
◦ Argument 2 – Y axis, positive non zero number

• cFZX_PARAMETER_ZPOSITION
◦ Sets the body render order, depends on renderer to implement.
◦ Argument 1 – Z axis 
◦ Argument 2 – unused 

• cFZX_PARAMETER_UV0
◦ Texture Coordinates, depends on renderer to implement.
◦ Argument 1 – X axis, positive non zero number 
◦ Argument 2 – Y axis, positive non zero number

• cFZX_PARAMETER_UV1
◦ Texture Coordinates, depends on renderer to implement.
◦ Argument 1 – X axis, positive non zero number 
◦ Argument 2 – Y axis, positive non zero number

• cFZX_PARAMETER_UV2
◦ Texture Coordinates, depends on renderer to implement.
◦ Argument 1 – X axis, positive non zero number 
◦ Argument 2 – Y axis, positive non zero number

• cFZX_PARAMETER_UV3
◦ Texture Coordinates, depends on renderer to implement.
◦ Argument 1 – X axis, positive non zero number 
◦ Argument 2 – Y axis, positive non zero number



Querying Body Parameters:

Making this easier is on the To-Do list. All of the parameters that have been set 
can be read by looking at the __fzxBody() structure. The structure is defined in 
the fzxNGN_ini.bas file. 

From the earlier example we can look at the current position

1. PRINT __fzxBody(box).fzx.position.x
2. PRINT __fzxBody(box).fzx.position.y



Appendix I: List of Subs and Functions

/fzxNGN/fzxNGN_BASE_v2/fzxNGN_AABB.bas
FUNCTION fzxAABBOverlap (Ax AS DOUBLE, Ay AS DOUBLE, Aw AS DOUBLE, Ah AS DOUBLE, Bx AS DOUBLE, By AS 
DOUBLE, Bw AS DOUBLE, Bh AS DOUBLE)
FUNCTION fzxAABBOverlapVector (A AS tFZX_VECTOR2d, Aw AS DOUBLE, Ah AS DOUBLE, B AS tFZX_VECTOR2d, Bw 
AS DOUBLE, Bh AS DOUBLE)
FUNCTION fzxAABBOverlapObjects (a AS LONG, b AS LONG)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_BASE.bas
SUB fzxCircleInitialize (index AS LONG)
SUB fzxCircleComputeMass (index AS LONG, density AS DOUBLE)
SUB fzxPolygonInitialize (index AS LONG)
SUB fzxPolygonComputeMass (index AS LONG, density AS DOUBLE)
FUNCTION fzxCreateCircleBody (index AS LONG, radius AS DOUBLE)
FUNCTION fzxCreateCircleBodyEx (objName AS STRING, radius AS DOUBLE)
FUNCTION fzxCreateBoxBody (index AS LONG, xs AS DOUBLE, ys AS DOUBLE)
FUNCTION fzxCreateBoxBodyEx (objName AS STRING, xs AS DOUBLE, ys AS DOUBLE)
SUB fzxCreateTrapBody (index AS LONG, xs AS DOUBLE, ys AS DOUBLE, yoff1 AS DOUBLE, yoff2 AS DOUBLE)
FUNCTION fzxCreateTrapBodyEx (objName AS STRING, xs AS DOUBLE, ys AS DOUBLE, yoff1 AS DOUBLE, yoff2 AS 
DOUBLE)
SUB fzxBodyCreate (index AS LONG, shape AS tFZX_SHAPE)
SUB fzxBodyCreateEx (objname AS STRING, shape AS tFZX_SHAPE, index AS LONG)
SUB fzxBoxCreate (index AS LONG, sizex AS DOUBLE, sizey AS DOUBLE)
SUB fzxTrapCreate (index AS LONG, sizex AS DOUBLE, sizey AS DOUBLE, yOff1 AS DOUBLE, yOff2 AS DOUBLE)
SUB fzxCreateTerrainBody (index AS LONG, slices AS LONG, sliceWidth AS DOUBLE, nominalHeight AS DOUBLE)
SUB fzxCreateTerrainBodyEx (objName AS STRING, elevation() AS DOUBLE, slices AS LONG, sliceWidth AS 
DOUBLE, nominalHeight AS DOUBLE)
SUB fzxTerrainCreate (index AS LONG, ele1 AS DOUBLE, ele2 AS DOUBLE, sliceWidth AS DOUBLE, 
nominalHeight AS DOUBLE)
SUB fzxVShapeCreate (index AS LONG, sizex AS DOUBLE, sizey AS DOUBLE)
SUB fzxVertexSet (index AS LONG, verts() AS tFZX_VECTOR2d)
SUB fzxBodyDelete (index AS LONG, perm AS _BYTE)
FUNCTION fzxFindRightMostVert (verts() AS tFZX_VECTOR2d)
SUB fzxVector2DGetSupport (index AS LONG, dir AS tFZX_VECTOR2d, bestVertex AS tFZX_VECTOR2d)
SUB fzxShapeCreate (sh AS tFZX_SHAPE, ty AS LONG, radius AS DOUBLE)
SUB fzxSetBody (Parameter AS LONG, Index AS LONG, arg1 AS DOUBLE, arg2 AS DOUBLE)
SUB fzxSetBodyEx (parameter AS LONG, objName AS STRING, arg1 AS DOUBLE, arg2 AS DOUBLE)
FUNCTION fzxGetBodyD# (Parameter AS LONG, Index AS LONG, arg AS _BYTE)
SUB fzxBodyStop (index AS LONG)
SUB fzxBodyOffset (index AS LONG, vec AS tFZX_VECTOR2d)
SUB fzxBodySetStatic (index AS LONG, arg AS LONG)
FUNCTION fzxBodyAtRest (index AS LONG, minVel AS DOUBLE)
SUB fzxCopyBodies (body() AS tFZX_BODY, newBody() AS tFZX_BODY)
FUNCTION fzxArrayNextIndex (i AS LONG, count AS LONG)
SUB fzxCollisionCCHandle (m AS tFZX_MANIFOLD, contacts() AS tFZX_VECTOR2d, A AS LONG, B AS LONG)
SUB fzxCollisionPCHandle (m AS tFZX_MANIFOLD, contacts() AS tFZX_VECTOR2d, A AS LONG, B AS LONG)
SUB fzxCollisionCPHandle (m AS tFZX_MANIFOLD, contacts() AS tFZX_VECTOR2d, A AS LONG, B AS LONG)
FUNCTION fzxCollisionPPClip (n AS tFZX_VECTOR2d, c AS DOUBLE, face() AS tFZX_VECTOR2d)
SUB fzxCollisionPPFindIncidentFace (v() AS tFZX_VECTOR2d, RefPoly AS LONG, IncPoly AS LONG, 
referenceIndex AS LONG)
SUB fzxCollisionPPHandle (m AS tFZX_MANIFOLD, contacts() AS tFZX_VECTOR2d, A AS LONG, B AS LONG)
FUNCTION fzxCollisionPPFindAxisLeastPenetration (faceIndex() AS LONG, A AS LONG, B AS LONG)
SUB fzxImpulseIntegrateForces (index AS LONG, dt AS DOUBLE)
SUB fzxImpulseIntegrateVelocity (index AS LONG, dt AS DOUBLE)
SUB fzxImpulseStep (dt AS DOUBLE, iterations AS LONG)
SUB fzxBodyApplyImpulse (index AS LONG, fzxImpulse AS tFZX_VECTOR2d, contactVector AS tFZX_VECTOR2d)
SUB fzxManifoldInit (m AS tFZX_MANIFOLD, contacts() AS tFZX_VECTOR2d)
SUB fzxManifoldApplyImpulse (m AS tFZX_MANIFOLD, contacts() AS tFZX_VECTOR2d)



SUB fzxManifoldPositionalCorrection (m AS tFZX_MANIFOLD)
SUB fzxManifoldInfiniteMassCorrection (A AS LONG, B AS LONG)
FUNCTION fzxJointCreate (b1 AS LONG, b2 AS LONG, x AS DOUBLE, y AS DOUBLE)
FUNCTION fzxJointCreateEx (b1 AS LONG, b2 AS LONG, anchor1 AS tFZX_VECTOR2d, anchor2 AS tFZX_VECTOR2d)
SUB fzxJointDelete (d AS LONG)
SUB fzxJointSet (index AS LONG, b1 AS LONG, b2 AS LONG, x AS DOUBLE, y AS DOUBLE)
SUB fzxJointSetEx (index AS LONG, b1 AS LONG, b2 AS LONG, anchor1 AS tFZX_VECTOR2d, anchor2 AS 
tFZX_VECTOR2d)
SUB fzxJointPrestep (index AS LONG, inv_dt AS DOUBLE)
SUB fzxJointApplyImpulse (index AS LONG)
FUNCTION fzxIsBodyTouchingBody (A AS LONG, B AS LONG)
FUNCTION fzxIsBodyTouchingStatic (A AS LONG)
FUNCTION fzxIsBodyTouching (A AS LONG)
FUNCTION fzxHighestCollisionVelocity (hits() AS tFZX_HIT, A AS LONG) ' this function is a bit dubious 
and may not do as you think
FUNCTION fzxBodyManagerAdd ()
FUNCTION fzxBodyWithHash (hash AS _INTEGER64)
FUNCTION fzxBodyWithHashMask (hash AS _INTEGER64, mask AS LONG)
FUNCTION fzxBodyManagerID (objName AS STRING)
FUNCTION fzxBodyContainsString (start AS LONG, s AS STRING)
FUNCTION fzxComputeHash&& (s AS STRING)
SUB fzxHandleNetwork (net AS tFZX_NETWORK)
SUB fzxNetworkStartHost (net AS tFZX_NETWORK)
SUB fzxNetworkReceiveFromHost (net AS tFZX_NETWORK)
SUB fzxNetworkTransmit (net AS tFZX_NETWORK)
SUB fzxNetworkClose (net AS tFZX_NETWORK)
SUB fzxFSMChangeState (fsm AS tFZX_FSM, newState AS LONG)
SUB fzxFSMChangeStateEx (fsm AS tFZX_FSM, newState AS LONG, arg1 AS tFZX_VECTOR2d, arg2 AS 
tFZX_VECTOR2d, arg3 AS LONG)
SUB fzxFSMChangeStateOnTimer (fsm AS tFZX_FSM, newstate AS LONG)
FUNCTION fzxReadArrayLong& (s AS STRING, p AS LONG)
SUB fzxSetArrayLong (s AS STRING, p AS LONG, v AS LONG)
FUNCTION fzxReadArraySingle! (s AS STRING, p AS LONG)
SUB fzxSetArraySingle (s AS STRING, p AS LONG, v AS SINGLE)
FUNCTION fzxReadArrayInteger% (s AS STRING, p AS LONG)
SUB fzxSetArrayInteger (s AS STRING, p AS LONG, v AS INTEGER)
FUNCTION fzxReadArrayDouble# (s AS STRING, p AS LONG)
SUB fzxSetArrayDouble (s AS STRING, p AS LONG, v AS DOUBLE)
SUB fzxInitFPS
SUB fzxFPS
SUB fzxFPSMain
SUB fzxFPSdt
SUB fzxHandleFPSMain
SUB fzxHandleFPSGL
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_CAMERA.bas
SUB fzxWorldToCamera (index AS INTEGER, vert AS tFZX_VECTOR2d)
SUB fzxWorldToCameraEx (posVert AS tFZX_VECTOR2d, vert AS tFZX_VECTOR2d)
SUB fzxCalculateFOV
SUB fzxCameraToWorld (oVec AS tFZX_VECTOR2d, iVec AS tFZX_VECTOR2d)
SUB fzxCameratoWorldEx (iVec AS tFZX_VECTOR2d, oVec AS tFZX_VECTOR2d)
SUB fzxCameratoWorldScEx (iVec AS tFZX_VECTOR2d, oVec AS tFZX_VECTOR2d)
SUB fzxWorldToCameraBody (index AS LONG, vert AS tFZX_VECTOR2d)
SUB fzxWorldToCameraBodyNR (index AS LONG, vert AS tFZX_VECTOR2d)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_GL.bas
SUB glDrawCircle (cx AS SINGLE, cy AS SINGLE, r AS SINGLE, numOfSegments AS LONG)
SUB glDrawRectText (tex AS LONG, x AS SINGLE, y AS SINGLE, x1 AS SINGLE, y1 AS SINGLE)
SUB glDrawTexturedQuad (tex AS LONG, a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d, d AS 
tFZX_VECTOR2d)
SUB glDrawLine (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS LONG, w AS INTEGER)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_IMPULSEMATH.bas



FUNCTION fzxImpulseEqual (a AS DOUBLE, b AS DOUBLE)
FUNCTION fzxImpulseClamp# (min AS DOUBLE, max AS DOUBLE, a AS DOUBLE)
FUNCTION fzxImpulseRound# (a AS DOUBLE)
FUNCTION fzxImpulseRandomFloat## (min AS _FLOAT, max AS _FLOAT)
FUNCTION fzxImpulseRandomInteger% (min AS INTEGER, max AS INTEGER)
FUNCTION fzxImpulseRandomdouble# (min AS DOUBLE, max AS DOUBLE)
FUNCTION fzxImpulseGT (a AS DOUBLE, b AS DOUBLE)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_INPUT.bas
SUB fzxHandleInputDevice
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_LINESEG.bas
SUB fzxLineIntersection (l1 AS tFZX_LINE2d, l2 AS tFZX_LINE2d, o AS tFZX_VECTOR2d)
FUNCTION fzxLineSegmentsIntersect (l1 AS tFZX_LINE2d, l2 AS tFZX_LINE2d)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_MATRIXMATH.bas
SUB fzxMatrix2x2SetRadians (m AS tFZX_MATRIX2D, radians AS DOUBLE)
SUB fzxMatrix2x2SetScalar (m AS tFZX_MATRIX2D, a AS DOUBLE, b AS DOUBLE, c AS DOUBLE, d AS DOUBLE)
SUB fzxMatrix2x2Abs (m AS tFZX_MATRIX2D, o AS tFZX_MATRIX2D)
SUB fzxMatrix2x2GetAxisX (m AS tFZX_MATRIX2D, o AS tFZX_VECTOR2d)
SUB fzxMatrix2x2GetAxisY (m AS tFZX_MATRIX2D, o AS tFZX_VECTOR2d)
SUB fzxMatrix2x2TransposeI (m AS tFZX_MATRIX2D)
SUB fzxMatrix2x2Transpose (m AS tFZX_MATRIX2D, o AS tFZX_MATRIX2D)
SUB fzxMatrix2x2Invert (m AS tFZX_MATRIX2D, o AS tFZX_MATRIX2D)
SUB fzxMatrix2x2MultiplyVector (m AS tFZX_MATRIX2D, v AS tFZX_VECTOR2d, o AS tFZX_VECTOR2d)
SUB fzxMatrix2x2AddMatrix (m AS tFZX_MATRIX2D, x AS tFZX_MATRIX2D, o AS tFZX_MATRIX2D)
SUB fzxMatrix2x2MultiplyMatrix (m AS tFZX_MATRIX2D, x AS tFZX_MATRIX2D, o AS tFZX_MATRIX2D)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_MEM.bas
SUB fzxSetBodyVertXY (indexBody AS LONG, indexVert AS LONG, x AS DOUBLE, y AS DOUBLE)
SUB fzxSetBodyVert (indexBody AS LONG, indexVert AS LONG, v AS tFZX_VECTOR2d)
FUNCTION fzxGetBodyVertX# (indexBody AS LONG, indexVert AS LONG)
FUNCTION fzxGetBodyVertY# (indexBody AS LONG, indexVert AS LONG)
SUB fzxGetBodyVert (indexBody AS LONG, indexVert AS LONG, vert AS tFZX_VECTOR2d)
SUB fzxSetBodyNormXY (indexBody AS LONG, indexNorm AS LONG, x AS DOUBLE, y AS DOUBLE)
SUB fzxSetBodyNorm (indexBody AS LONG, indexNorm AS LONG, v AS tFZX_VECTOR2d)
FUNCTION fzxGetBodyNormX# (indexBody AS LONG, indexNorm AS LONG)
FUNCTION fzxGetBodyNormY# (indexBody AS LONG, indexNorm AS LONG)
SUB fzxGetBodyNorm (indexBody AS LONG, indexNorm AS LONG, norm AS tFZX_VECTOR2d)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_PERLIN.bas
FUNCTION fzxPerlinScaleOffset (p AS DOUBLE)
FUNCTION fzxPerlinInterpolate# (a0 AS DOUBLE, a1 AS DOUBLE, w AS DOUBLE)
SUB fzxPerlinRandomGradient (seed AS DOUBLE, ix AS INTEGER, iy AS INTEGER, o AS tFZX_VECTOR2d)
FUNCTION fzxPerlinDotGridGradient# (seed AS DOUBLE, ix AS INTEGER, iy AS INTEGER, x AS DOUBLE, y AS 
DOUBLE)
FUNCTION fzxPerlin# (x AS DOUBLE, y AS DOUBLE, seed AS DOUBLE)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_POLYGON.bas
SUB fzxPolygonMakeCCW (obj AS tFZX_TRIANGLE)
FUNCTION fzxPolygonIsReflex (t AS tFZX_TRIANGLE)
SUB fzxPolygonSetOrient (index AS LONG, radians AS DOUBLE)
SUB fzxPolygonInvertNormals (index AS LONG)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_SCALARMATH.bas
FUNCTION fzxScalarMin# (a AS DOUBLE, b AS DOUBLE)
FUNCTION fzxScalarMax# (a AS DOUBLE, b AS DOUBLE)
FUNCTION fzxScalarMap# (x AS DOUBLE, in_min AS DOUBLE, in_max AS DOUBLE, out_min AS DOUBLE, out_max AS 
DOUBLE)
FUNCTION fzxScalarLERP# (current AS DOUBLE, target AS DOUBLE, t AS DOUBLE)
FUNCTION fzxScalarLERPSmooth# (current AS DOUBLE, target AS DOUBLE, t AS DOUBLE)
FUNCTION fzxScalarLERPSmoother# (current AS DOUBLE, target AS DOUBLE, t AS DOUBLE)
FUNCTION fzxScalarLERPProgress# (startTime AS DOUBLE, endTime AS DOUBLE)
FUNCTION fzxScalarRoughEqual (a AS DOUBLE, b AS DOUBLE, tolerance AS DOUBLE)
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_VECMATH.bas
SUB fzxVector2DSet (v AS tFZX_VECTOR2d, x AS DOUBLE, y AS DOUBLE)
SUB fzxVector2dSetVector (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d)



SUB fzxVector2dNeg (v AS tFZX_VECTOR2d)
SUB fzxVector2DNegND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d)
SUB fzxVector2DMultiplyScalar (v AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DMultiplyScalarND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DDivideScalar (v AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DDivideScalarND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DAddScalar (v AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DAddScalarND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DMultiplyVector (v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DMultiplyVectorND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DDivideVector (v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DDivideVectorND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DAddVector (v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DAddVectorND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DAddVectorScalar (v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d, s AS DOUBLE)
SUB fzxVector2DAddVectorScalarND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d, s AS 
DOUBLE)
SUB fzxVector2DSubVector (v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DSubVectorND (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, m AS tFZX_VECTOR2d)
SUB fzxVector2DSwap (v1 AS tFZX_VECTOR2d, v2 AS tFZX_VECTOR2d)
FUNCTION fzxVector2DLengthSq# (v AS tFZX_VECTOR2d)
FUNCTION fzxVector2DLength# (v AS tFZX_VECTOR2d)
SUB fzxVector2DRotate (v AS tFZX_VECTOR2d, radians AS DOUBLE)
SUB fzxVector2DNormalize (v AS tFZX_VECTOR2d)
SUB fzxVector2DMin (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, o AS tFZX_VECTOR2d)
SUB fzxVector2DMax (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, o AS tFZX_VECTOR2d)
FUNCTION fzxVector2DDot# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d)
FUNCTION fzxVector2DSqDist# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d)
FUNCTION fzxVector2DDistance# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d)
FUNCTION fzxVector2DCross# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d)
SUB fzxVector2DCrossScalar (o AS tFZX_VECTOR2d, v AS tFZX_VECTOR2d, a AS DOUBLE)
FUNCTION fzxVector2DArea# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d)
FUNCTION fzxVector2DLeft# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d)
FUNCTION fzxVector2DLeftOn# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d)
FUNCTION fzxVector2DRight# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d)
FUNCTION fzxVector2DRightOn# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d)
FUNCTION fzxVector2DCollinear# (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, c AS tFZX_VECTOR2d, 
thresholdAngle AS DOUBLE)
SUB fzxVector2DLERP (curr AS tFZX_VECTOR2d, start AS tFZX_VECTOR2d, target AS tFZX_VECTOR2d, inc AS 
DOUBLE)
SUB fzxVector2DLERPSmooth (curr AS tFZX_VECTOR2d, start AS tFZX_VECTOR2d, target AS tFZX_VECTOR2d, inc 
AS DOUBLE)
SUB fzxVector2DLERPSmoother (curr AS tFZX_VECTOR2d, start AS tFZX_VECTOR2d, target AS tFZX_VECTOR2d, 
inc AS DOUBLE)
SUB fzxVector2DOrbitVector (o AS tFZX_VECTOR2d, position AS tFZX_VECTOR2d, dist AS DOUBLE, angle AS 
DOUBLE)
FUNCTION fzxVector2DEqual (a AS tFZX_VECTOR2d, b AS tFZX_VECTOR2d, tolerance AS DOUBLE)
SUB fzxVector2DMid (o AS tFZX_VECTOR2d, v1 AS tFZX_VECTOR2d, v2 AS tFZX_VECTOR2d)
FUNCTION fzxGetAngleVector2d# (p1 AS tFZX_VECTOR2d, p2 AS tFZX_VECTOR2d)
FUNCTION fzxGetAngle# (x1 AS DOUBLE, y1 AS DOUBLE, x2 AS DOUBLE, y2 AS DOUBLE) 'returns 0-359.99...
/fzxNGN/fzxNGN_BASE_v2/fzxNGN_XML.bas
SUB XMLparse (file AS STRING, con() AS tFZX_STRINGTUPLE)
FUNCTION getXMLArgValue# (i AS STRING, s AS STRING)
FUNCTION getXMLArgString$ (i AS STRING, s AS STRING)
FUNCTION topStackString$ (stack() AS STRING)
SUB pushStackString (stack() AS STRING, element AS STRING)
SUB popStackString (stack() AS STRING)
SUB pushStackContextArg (stack() AS tFZX_STRINGTUPLE, element_name AS STRING, element AS STRING)
SUB pushStackContext (stack() AS tFZX_STRINGTUPLE, element AS tFZX_STRINGTUPLE)
SUB popStackContext (stack() AS tFZX_STRINGTUPLE)



SUB pushStackVector (stack() AS tFZX_VECTOR2d, element AS tFZX_VECTOR2d)
SUB popStackVector (stack() AS tFZX_VECTOR2d)
SUB topStackVector (o AS tFZX_VECTOR2d, stack() AS tFZX_VECTOR2d)
SUB loadFilebyLine (fl AS STRING, filetext() AS STRING)
FUNCTION trim$ (s AS STRING)
FUNCTION isAlpha (c AS STRING)
FUNCTION isDigit (c AS STRING)
FUNCTION isSymbol (c AS STRING)
/fzxNGN/fzxNGN_BASE_v2/subfuncLister.bas
SUB OSDirectoryListing (os AS tOS, directory AS STRING, file() AS STRING * 256)
SUB OSinit (os AS tOS)


